
Rozdział 6.

Odwołania do elementów
lub fragmentów tablic

Użycie funkcji INDEKS
W przypadku, gdy nie potrzeba ujawniać całej tablicy wyników funkcji czy operacji tablicowej,
a jedynie wybrane elementy wynikowej tablicy, należy użyć funkcji INDEKS. Umożliwia ona wybór
pojedynczego elementu lub wskazanych elementów tablicy danych lub wyników. Jeżeli funkcja
ma zwrócić pojedynczą wartość, zatwierdza się ją samym klawiszem Enter, jeżeli zaś wynikiem
ma być tablica lub obszar, zaznacza się najpierw miejsce na wynik, a następnie zatwierdza przez
kombinację Ctrl+Shift+Enter.

Funkcja INDEKS ma formalnie (według dokumentacji) dwa warianty składni: tablicową i adresową.
W zależności od tego, co jest pierwszym argumentem, może mieć zastosowanie forma tablicowa:

INDEKS(tablica; nr_wiersza; nr_kolumny)

lub forma adresowa, zwana też odwołaniową, która różni się tylko obecnością dodatkowego opcjo-
nalnego argumentu:

INDEKS(odwołanie; nr_wiersza; nr_kolumny; nr_obszaru)

Pierwszy argument funkcji INDEKS będę nazywał argumentem głównym, pozostałe — argu-
mentami indeksowymi.

Tablica jest tablicą albo wyrażeniem dającym w wyniku tablicę, a odwołanie jest zakresem ko-
mórek, który może obejmować więcej niż jeden obszar. Jeśli odwołanie obejmuje adresy kilku
obszarów, ich listę należy ująć w nawiasy. W tym miejscu może wystąpić również wyrażenie,
dające w wyniku odwołanie.

Nr_wiersza wybiera wiersz tablicy lub obszaru, z którego ma pochodzić wartość wynikowa. Jeśli
pominiemy ten argument lub zostanie użyta wartość 0, to w zależności od sposobu użycia funk-
cji wybrane zostaną albo wszystkie wiersze, albo wiersz bieżący, albo pierwszy.

Argument nr_kolumny określa kolumnę tablicy lub obszaru, z której ma pochodzić wartość wyniko-
wa. Jeśli pominiemy go lub zostanie użyta wartość 0, to w zależności od sposobu użycia funkcji wy-
brane zostaną albo wszystkie kolumny, albo kolumna bieżąca, albo pierwsza.

Nr_wiersza i nr_kolumny muszą odnosić się do elementu lub komórki wewnątrz tablicy lub obszaru,
w innym przypadku funkcja INDEKS podaje w wyniku wartość błędu #ADR!, ale tylko w tych
wierszach lub kolumnach, które zostały błędnie zaadresowane. Wiersze i kolumny argumentu
głównego są numerowane od 1. Jeżeli odwołanie obejmuje kilka obszarów, numeracja każdego
z nich jest niezależna.

Jeśli zostanie użyty zarówno argument nr_wiersza, jak i nr_kolumny, to funkcja INDEKS daje w wyni-
ku odwołanie do komórki na przecięciu wiersza i kolumny o numerach podanych w nr_wiersza
i nr_kolumny lub odpowiednią wartość z tablicy.

Jeśli argument główny zawiera tylko jeden wiersz lub kolumnę, to jeden z argumentów nr_wiersza
lub nr_kolumny (wraz z poprzedzającym średnikiem) można pominąć. Można również pozostawić
sam średnik lub użyć argumentu o wartości 0 lub 1.

Jeśli argumentem głównym funkcji INDEKS jest odwołanie do obszaru dwuwymiarowego lub za-
kresu nieregularnego, aby wyrażenie było poprawne składniowo, w wywołaniu muszą być co
najmniej dwa średniki; w innych przypadkach wystarczy jeden.

Argument nr_obszaru określa, z którego obszaru ma zostać zwrócony wynik. Ma on znaczenie,
gdy odwołanie obejmuje kilka obszarów; w razie pominięcia tego argumentu (wraz z poprzedzają-
cym średnikiem) uwzględniony zostanie pierwszy obszar. Argument ten nie może mieć wartości ze-
rowej, nie można też pozostawić w wywołaniu trzeciego średnika bez wartości. Jeśli argumen-
tem głównym jest odwołanie do pojedynczego obszaru lub tablica, to argument nr_obszaru
(jeśli wystąpi) musi mieć wartość 1. Wynika stąd, że w istocie nie ma dwóch wariantów składni,
a tylko jeden — wariant odwołaniowy z ograniczeniami dotyczącymi argumentu nr_obszaru.

Jeśli argument główny jest tablicą, funkcja zwraca w wyniku zawsze pojedynczą wartość lub ta-
blicę, jeśli zaś ten argument jest odwołaniem, funkcja zwraca odwołanie, o ile jest to możliwe,
a jeśli nie — zwraca wartość lub tablicę wartości.

Argumenty indeksowe mogą mieć również formę tablic lub zakresów komórek zawierających
numery wierszy, kolumn lub obszarów. Ta funkcjonalność nie jest jednak opisana w oficjalnej
dokumentacji. Należy jej używać z dużą ostrożnością, szczególnie gdy formuły są złożone, bo wynik
nie zawsze da się łatwo przewidzieć [B20]. Właśnie w takich przypadkach funkcja INDEKS po-
mimo użycia formy adresowej zwraca tablice, a nie odwołania.

Argumentem głównym funkcji INDEKS nie mogą być tzw. odwołania 3-W (piszę o nich w roz-
dziale 4., w podrozdziale „Użycie i interpretacja odwołań wielopoziomowych”), gdyż w sensie for-
malnym nie są to ani odwołania, ani tablice.

Argument główny powinien być tablicą lub odwołaniem
Jeżeli jako argumentu głównego funkcji INDEKS użyto stałej lub wyrażenia zwracającego pojedynczą
wartość, wynik jest poprawny tylko wtedy, gdy jest to liczba. W każdym innym przypadku
funkcja zwraca wartość błędu (jeżeli wyrażenie zwraca błąd, otrzymamy wartość tego błędu, je-
śli zaś wyrażenie zwraca wartość nieliczbową, wynikiem będzie wartość błędu #ARG!). Przyczy-
ną takiego zachowania funkcji INDEKS jest inna — w porównaniu z pojedynczą wartością —
struktura tablicy. Aby uniknąć tego problemu (kiedy tablica może czasem być pojedynczą warto-
ścią nieliczbową), należy zastosować konstrukcję:

= INDEKS(JEŻELI({1};tablica); nr_wiersza; nr_kolumny)

Opisana wyżej sytuacja dotyczy również innych funkcji tablicowych.

Użycie zera lub pominięcie wartości indeksu
Jeżeli wykorzystany zostanie tylko jeden argument: nr_wiersza lub nr_kolumny, a drugi zostanie
pominięty (gdy argumentem jest odwołanie do zakresu dwuwymiarowego, średnik musi pozostać)
lub jako argument indeksowy zostanie podana wartość 0, to INDEKS da w wyniku odwołanie lub
tablicę z całym wierszem lub kolumną. Aby wyświetlić te wartości w arkuszu jako sekwencję,
należy wprowadzić INDEKS jako funkcję tablicową, naciskając Ctrl+Shift+Enter. Przy zatwierdzeniu
samym klawiszem Enter, jeśli argumentem głównym jest odwołanie, funkcja INDEKS zwraca
wartość z tego samego wiersza lub tej samej kolumny na zasadzie rzutu prostokątnego, analo-
gicznie jak w przypadku odwołania do zakresu w zwykłej formule, natomiast jeśli argumentem
głównym jest tablica, INDEKS zwraca jej pierwszy element (por. rozdział 7., punkt „Dane gru-
powe w zwykłych wyrażeniach”). Opis ten dotyczy typowego użycia funkcji INDEKS, to znaczy
takiego, w którym argumenty indeksowe nie mają struktury tablicy, lecz są zwykłymi liczbami.

Użycie funkcji INDEKS do obszaru dwuwymiarowego
Przykład 6.1.
Jako źródło danych wykorzystamy zakres (obszar) TA o wymiarach 4×4, umieszczony w arkuszu.

1 2 4 6

3 5 –3 7

6 8 10 –2

5 7 4 3

Jego pierwszą kolumnę oznaczymy TX, a pierwszy wiersz — TY.

Odwołanie do drugiego elementu (wiersza) tablicy TX może mieć jedną z postaci:

= INDEKS(TX;2)
= INDEKS(TX;2;)
= INDEKS(TX;2;0)
= INDEKS(TX;2;1)

Podobnie, odwołanie do trzeciego elementu (kolumny) tablicy TY może mieć postać (alterna-
tywnie):

= INDEKS(TY;3)
= INDEKS(TY;;3)
= INDEKS(TY;0;3)
= INDEKS(TY;1;3)

Jeśli TA oznacza tablicę lub zakres dwuwymiarowy, możliwe są na przykład następujące odwo-
łania tablicowe:

= INDEKS(TA;1;)

= INDEKS(TA;1;0) — w obu przypadkach wynikiem jest cały pierwszy wiersz tablicy TA.

Jeśli TA jest odwołaniem do zakresu, wyrażenie = INDEKS(TA;1) byłoby błędne, ale już wyrażenie
= INDEKS(+TA;1) jest poprawne. W przypadku, gdy TA oznacza stałą tablicową, prawidłowe są oba
ostatnie wyrażenia.

= INDEKS(TA;;2)

= INDEKS(TA;0;2) — w obu przypadkach wynikiem jest cała druga kolumna tablicy TA.

Jeśli TA jest odwołaniem do zakresu, wyrażenie = INDEKS(TA;2) byłoby błędne.

Pojedynczy element tablicy lub obszaru TA odczytamy za pomocą wyrażenia:

= INDEKS(TA;2;4)

W tym przypadku będzie to element na przecięciu drugiego wiersza i czwartej kolumny (war-
tość 7 lub odwołanie do komórki zawierającej tę wartość).

Odwołanie do komórki o wyliczonym adresie
Pierwszy argument funkcji INDEKS może reprezentować cały arkusz, co można zapisać jako za-
kres wierszy $1:$1048576 (od wersji 2007) bądź $1:$65536 (we wcześniejszych wersjach). Jeżeli
odwołanie ma dotyczyć arkusza innego niż bieżący, należy je poprzedzić kwalifikatorem arkusza.
Można oczywiście użyć również mniejszego zakresu wierszy lub odpowiedniego zakresu ko-
lumn, począwszy od kolumny A. Użycie funkcji INDEKS w takiej formie umożliwia odwołanie do
komórki lub komórek o wyliczonym adresie. Jeśli nie pamiętamy numeru ostatniego wiersza
lub ostatniej kolumny, to wszystko można wyliczyć:
 numer ostatniego wiersza — = ILE.WIERSZY(A:A)
 numer ostatniej kolumny — = LICZBA.KOLUMN(1:1)
 oznaczenie literowe ostatniej kolumny — = PODSTAW(ADRES(1;LICZBA.KOLUMN(1:1);4);1;"")

Aby odczytać wartość z komórki o określonych absolutnych numerach wiersza i kolumny, można
użyć wyrażenia:

= INDEKS(ADR.POŚR("nazwa_arkusza!1:"&ILE.WIERSZY($A:$A)); numer_wiersza; numer_kolumny)

Ten sam wynik można uzyskać za pomocą funkcji ADR.POŚR i ADRES:

= ADR.POŚR(ADRES(numer_wiersza; numer_kolumny;;; nazwa_arkusza))

Jeżeli odwołanie dotyczy bieżącego arkusza, w obu wersjach można pominąć nazwę_arkusza.

Łączenie fragmentów tablic
Przykład 6.2.
Załóżmy, że mamy dwie tablice (lub obszary danych): tabl1, obejmującą trzy kolumny, oraz tabl2,
złożoną z dwóch kolumn. Za pomocą odpowiedniej formuły tablicowej w połączeniu z funkcją
INDEKS możemy utworzyć tablicę złożoną z pięciu kolumn, tak jakby były zapisane obok siebie.

= {1;0;0;0;0}*INDEKS(tabl1;;1)+ {0;1;0;0;0}*INDEKS(tabl1;;2)+
{0;0;1;0;0}*INDEKS(tabl1;;3)+ {0;0;0;1;0}*INDEKS(tabl2;;1)+
{0;0;0;0;1}*INDEKS(tabl2;;2)

Ta formuła jest analogiczna do użytej w przykładzie 4.2. Jej koncepcja polega na podziale ory-
ginalnych tablic na kolumny za pomocą funkcji INDEKS i połączeniu tak otrzymanych kolumn
za pomocą operacji tablicowych. Ze względu na zasadę superpozycji, która umożliwia dopasowanie
wymiarów przez kopiowanie tylko pojedynczych wierszy lub kolumn, przy takiej konstrukcji
wyrażenia nie ma możliwości uwzględnienia tablicy dwuwymiarowej w całości. Właściwości tej
formuły są podobne do tych z przykładu 4.2.: tablice muszą zawierać tylko dane liczbowe, for-
muła wymaga zatwierdzenia tablicowego, ale po takim zatwierdzeniu może być użyta do dal-
szych obliczeń. Jest to wyrażenie z odwołaniem lub tablicą.

Wygodniejsze okazuje się w tym przypadku użycie funkcji WYBIERZ:

= WYBIERZ({1;2;3;4;5}; INDEKS(tabl1;;1); INDEKS(tabl1;;2); INDEKS(tabl1;;3);
INDEKS(tabl2;;1); INDEKS(tabl2;;2))

albo jeszcze prościej, ale w sposób mniej oczywisty:

= WYBIERZ({1;1;1;2;3}; tabl1; INDEKS(tabl2;;1); INDEKS(tabl2;;2))

Czytelnikowi pozostawiam odpowiedź na pytanie, dlaczego tablicy tabl2 nie można przywołać
w jednej pozycji. Formuły z funkcją WYBIERZ dają wyniki niemal pełnowartościowe (TPA —
wyjaśnienie tego skrótu można znaleźć w rozdziale 4., w punkcie „Interpretacja argumentów gru-
powych przez funkcje JEŻELI i WYBIERZ”), gotowe do dalszego przetwarzania.

Alternatywne sposoby rozwiązania tego problemu korzystają z funkcji INDEKS w sposób niestandar-
dowy [B20]. Jeszcze inną metodę, z wykorzystaniem funkcji MACIERZ.ILOCZYN, pokazałem nieco da-
lej, w przykładzie 6.5. Problem można rozwiązać również posługując się Visual Basiciem [C6].

Zastosowania odwołaniowej formy funkcji INDEKS
Odwołaniowa forma funkcji INDEKS ma szczególne zastosowanie w przypadku zakresów o nie-
regularnych kształtach, złożonych z kilku obszarów. Jeżeli zakres złożony stanowią pojedyncze
komórki, można pominąć nr_wiersza i nr_kolumny i podać tylko nr_obszaru. W ten sposób można
zebrać rozproszone dane w jednym miejscu.

Wykorzystując fakt, że ta forma funkcji INDEKS daje w wyniku adres, można jej użyć do utwo-
rzenia zakresu dynamicznego o zmiennym kształcie:

adr_pocz:INDEKS(kom_kon;;;nr_obszaru)

Zakres kom_kon powinien być zakresem nieregularnym i składać się z komórek wyznaczających
prawe dolne narożniki konstruowanych obszarów. Początek zakresu jest w tym przykładzie stały,
ale można wyobrazić sobie, że w tym miejscu zamiast adresu podanego wprost użyjemy for-
muły zwracającej adres. Argument nr_obszaru wyznacza bieżący adres końca obszaru.

Analogiczny efekt można uzyskać za pomocą funkcji WYBIERZ z adresami komórek podanymi
jako argumenty. Przykład takiego rozwiązania podałem w rozdziale 4., w podrozdziale „Repre-
zentacja tablic w Excelu”.

Odwracanie porządku elementów w tablicy
Jeszcze innym przykładem użycia funkcji INDEKS jest odwracanie porządku elementów w tablicy.
Twórcy Excela nie przewidzieli specjalnej funkcji, która by rozwiązywała ten problem, na wzór
funkcji TRANSPONUJ. Załóżmy, że mamy zakres wierszowy, któremu nadamy nazwę zakres. Chcemy
wygenerować zakres o identycznych wymiarach, w którym kolejność elementów będzie odwrócona.

Kolejne komórki zawierające wynik opisane będą formułą zwykłą:

= INDEKS(zakres; LICZBA.KOLUMN(zakres)-NR.KOLUMNY(A1)+1)

LICZBA.KOLUMN(zakres) oznacza liczbę komórek w tym zakresie, NR.KOLUMNY(A1) zaś numery kolej-
nych kolumn, rozpoczynając od pierwszej. W pierwszej komórce zakresu wynikowego w wyrażeniu
NR.KOLUMNY(A1) musimy użyć adresu odnoszącego się do kolumny A; po skopiowaniu w prawo
adresy będą odnosić się do kolejnych kolumn, zapewniając otrzymanie oczekiwanego wyniku.

Gdybyśmy chcieli osiągnąć podobny efekt za pomocą formuły tablicowej i uzyskać od razu (jedną
formułą) całą tablicę wynikową, musielibyśmy użyć funkcji INDEKS w sposób niestandardowy [B20].

Uproszczenie zapisu przy wielokrotnym zagnieżdżeniu
funkcji JEŻELI
Gdy mamy do sprawdzenia kilka warunków i chcemy użyć wartości skojarzonej z pierwszym,
który ma wartość PRAWDA, możemy wykorzystać zagnieżdżoną funkcję JEŻELI:

= JEŻELI(war1; wartość1; JEŻELI(war2; wartość2; JEŻELI(war3; wartość3; wartość0)))

wartość0 jest wartością domyślną, której chcemy użyć, gdy żaden z warunków nie został spełniony.
W Excelu 2016+ podobne zadanie realizuje funkcja WARUNKI

= WARUNKI(war1; wartość1; war2; wartość2; war3; wartość3; PRAWDA; wartość0)

Każdy z tych zapisów ma jednak tę wadę, że wymaga modyfikacji przy zmianie liczby warun-
ków, a ponadto przy dużej ich liczbie staje się nieczytelny.

Jeżeli konstrukcja warunków i powiązanych wartości jest podobna i uda się je zapisać w postaci
ogólnych formuł tablicowych, powyższe zapisy można uprościć, wykorzystując funkcje INDEKS
i PODAJ.POZYCJĘ:

= JEŻELI.BŁĄD(INDEKS(wartości; PODAJ.POZYCJĘ(PRAWDA; warunki; 0)); wartość0)

Listowanie zawartości tablicy jednowymiarowej
wierszami lub kolumnami
Często otrzymujemy jako wynik jakichś obliczeń tablicę (najczęściej jednowymiarową) o nie-
znanej długości. Aby ją wyświetlić w całości, powinniśmy znać jej długość, w celu zarezerwowania
odpowiedniej liczby komórek na wynik. Alternatywą jest wyświetlanie elementów tablicy w kolej-
nych komórkach na podstawie formuły jednokomórkowej, kopiowanej tak długo, jak potrzeba.
Jeśli tablica jest jednowymiarowa, jej orientacja jest bez znaczenia. Możemy wyświetlać wyniki
w kolumnie, używając w formule funkcji ILE.WIERSZY, lub w wierszu, za pomocą formuły z funkcją
LICZBA.KOLUMN. Odpowiednie polecenia mają postać:

= JEŻELI.BŁĄD(INDEKS(tablica; ILE.WIERSZY(H$3:H3));"")
= JEŻELI.BŁĄD(INDEKS(tablica; LICZBA.KOLUMN($H3:H3));"")

H3 jest przykładowym adresem pierwszej komórki, w której chcemy ulokować początek listy.
Używamy tu adresu absolutnego lub mieszanego jako początkowego i adresu względnego jako
końcowego. Są to formuły zwykłe, gdyby jednak tablica była wyrażeniem złożonym, może być
wymagane zatwierdzenie tablicowe.

Użycie funkcji PRZESUNIĘCIE
Druga funkcja, która umożliwia odwoływanie się do fragmentów obszarów, nosi nazwę
PRZESUNIĘCIE. Pozwala ona na przykład „wyciąć” z obszaru dowolny fragment o kształcie pro-
stokąta (w tym pojedynczy wiersz lub kolumnę), a także odczytać zawartość obszaru położone-
go w określonej odległości od znanej komórki. Ściślej, funkcja zwraca odwołanie do komórki lub
obszaru, przesuniętych o podaną liczbę wierszy lub kolumn w stosunku do komórki odniesie-
nia. Funkcja PRZESUNIĘCIE spełnia podobne zadania, co INDEKS, ale można wskazać kilka istot-
nych różnic:
 funkcja PRZESUNIĘCIE przyjmuje jako argument główny tylko odwołania do obszarów,

podczas gdy INDEKS może przyjmować zarówno odwołania (i to również do zakresów
nieregularnych), jak i tablice;

 funkcja PRZESUNIĘCIE może zwracać wynik również spoza argumentu głównego,
natomiast INDEKS tylko w jego obrębie;

 INDEKS sprawdza się lepiej, gdy chcemy uzyskać w wyniku pojedyncze wartości lub całe
wiersze i całe kolumny, PRZESUNIĘCIE dobrze sobie radzi z uzyskaniem pojedynczych
wartości oraz dowolnych spójnych obszarów. Żadna z nich nie radzi sobie natomiast
dobrze z wybieraniem różnych fragmentów danych i łączeniem ich w nową całość;

 funkcja INDEKS numeruje wiersze i kolumny od 1, a PRZESUNIĘCIE — od 0. Należy o tym
pamiętać przy przekształcaniu wzorów wykorzystujących te funkcje;

 PRZESUNIĘCIE jest funkcją ulotną, co wpływa negatywnie na szybkość obliczeń, podczas
gdy INDEKS jest nieulotna.

Oto składnia funkcji PRZESUNIĘCIE:

PRZESUNIĘCIE(zakres_odniesienia; wiersze; kolumny; wysokość; szerokość)

Zakres_odniesienia to adres komórki (odwołanie) lub obszaru sąsiednich komórek, od którego
wyznacza się przesunięcie. Jeśli nie jest to pojedyncza komórka, przesunięcie liczy się od komórki
odniesienia, czyli od początku (lewego górnego rogu) tego obszaru. Jeżeli ten argument nie
oznacza odwołania, funkcja PRZESUNIĘCIE zwróci wartość błędu #ARG!.

Wiersze to liczba wierszy w górę (liczba ujemna) lub w dół (liczba dodatnia) liczonych od początku
zakresu_odniesienia, określająca położenie początku zakresu wynikowego. Argument wiersze
równy 0 oznacza ten sam wiersz. Liczbę 0 można pominąć, ale trzeba pozostawić poprzedzający
średnik.

Kolumny to liczba kolumn w lewo (liczba ujemna) lub w prawo (liczba dodatnia) liczonych od po-
czątku zakresu_odniesienia, określająca położenie początku zakresu wynikowego. Argument
kolumny równy 0 oznacza tę samą kolumnę. Liczbę 0 można pominąć, ale trzeba pozostawić po-
przedzający średnik.

Argumenty wiersze i kolumny będę określał mianem argumentów indeksowych.

Jeśli argumenty indeksowe przesuwają odwołanie poza brzeg arkusza, to funkcja PRZESUNIĘCIE daje
w wyniku wartość błędu #ADR! — i to we wszystkich komórkach wyniku, a nie tylko w tych,
które powodują błąd (inaczej niż w przypadku funkcji INDEKS). Formalnie jest to jedna wartość
błędu, która powiela się przy zatwierdzeniu w kilku komórkach.

Podając wartości opcjonalnych argumentów wysokość i szerokość, można określić rozmiary zakre-
su wynikowego, inne niż rozmiary zakresu_odniesienia.

Wysokość to liczba wierszy określająca wysokość zakresu wynikowego. Powinna być liczbą dodatnią.
Jeśli jest liczbą ujemną, funkcja zwraca zakres położony powyżej nowego zakresu_odniesienia,
ale bez zmiany kolejności komórek (jest to właściwość nieudokumentowana).

Szerokość to liczba kolumn określająca szerokość zakresu wynikowego. Także ona powinna być
liczbą dodatnią. Jeśli jest liczbą ujemną, funkcja zwraca zakres położony na lewo od nowego
zakresu_odniesienia, ale bez zmiany kolejności komórek (jest to właściwość nieudokumentowana).

Jeśli argumenty wysokość lub szerokość zostaną pominięte, to zakres wynikowy będzie mieć ta-
ką samą wysokość lub szerokość jak zakres_odniesienia. I to jest jedyna okoliczność, gdy roz-
miary zakresu_odniesienia mają znaczenie. Jeżeli wysokość i szerokość są podane jawnie, zna-
czenie ma tylko początek zakresu_odniesienia (komórka odniesienia).

Funkcja PRZESUNIĘCIE w rzeczywistości nie przesuwa żadnych komórek ani nie zmienia wyboru
zakresu, tylko zwraca odwołanie do zakresu wynikowego, który jest przesunięty w stosunku do
zakresu_odniesienia. W związku z tym może być wykorzystywana z innymi funkcjami wyma-
gającymi podania odwołania jako argumentu.

Jeżeli w zakresie działania funkcji PRZESUNIĘCIE wstawiamy lub usuwany wiersze lub kolumny,
wszystkie argumenty będące adresami są stosownie modyfikowane, natomiast stałe liczbowe
nie ulegają zmianie. Należy o tym pamiętać, gdyż w konsekwencji tego wstawienie lub usunię-
cie wiersza czy kolumny może zmienić wynik funkcji.

Podsumowując, gdy początek zakresu_odniesienia (komórkę odniesienia) oznaczymy symbolem
pzo, zakresy zwracane przez funkcję PRZESUNIĘCIE można obliczyć ze wzorów:

Gdy wysokość >0, zakres wierszy obejmuje

od WIERSZ(pzo)+wiersze
do WIERSZ(pzo)+wiersze+wysokość-1

Gdy wysokość <0, zakres wierszy obejmuje

od WIERSZ(pzo)+wiersze+wysokość+1
do WIERSZ(pzo)+wiersze

Gdy szerokość >0, zakres kolumn obejmuje

od NR.KOLUMNY(pzo)+kolumny
do NR.KOLUMNY(pzo)+kolumny+szerokość-1

Gdy szerokość <0, zakres kolumn obejmuje

od NR.KOLUMNY(pzo)+kolumny+szerokość+1
do NR.KOLUMNY(pzo)+kolumny

Można również zauważyć, że wartości argumentów wysokość lub szerokość równe –1 mają takie
samo znaczenie jak 1, a wartość 0 powoduje błąd #ADR!.

Użycie ujemnych wartości argumentów wysokość i szerokość, jakkolwiek dopuszczalne, nie po-
szerza możliwości funkcji, chociaż w pewnych sytuacjach pozwala skrócić zapis wyrażenia. Taki
sam wynik uzyskamy zmieniając wartości wysokości i szerokości na ich wartości bezwzględne
i równocześnie korygując odpowiednio argumenty wiersze i kolumny. Nowe wartości tych ar-
gumentów to:

wiersze-MODUŁ.LICZBY(wysokość)+1 oraz kolumny-MODUŁ.LICZBY(szerokość)+1

Gdy funkcja PRZESUNIĘCIE jest wywoływana w formule zwykłej, a wynik obejmuje formalnie kilka
komórek, efektywny wynik jest wyznaczany według zasady rzutu prostokątnego, ale w odnie-
sieniu do wyniku formuły, a nie do adresu argumentu (por. rozdział 2., podpunkt „Interpreta-
cja adresów obszarów w zwykłych wyrażeniach”). Przykładowo wywołanie

= PRZESUNIĘCIE($B$2:$B$4;1;1)

powinno dać w wyniku zakres $C$3:$C$5. Jeśli to wywołanie umieścimy w dowolnej kolumnie
w trzecim wierszu, otrzymamy w wyniku zawartość komórki $C$3, a w czwartym — zawartość
komórki $C$4. W każdym wierszu spoza zakresu 3:5 zwrócona zostanie wartość błędu #ARG!.
Jeżeli natomiast podamy wysokość równą 1, to w dowolnej komórce powyższe wywołanie będzie
odnosić się do komórki $C$3.

Należy pamiętać, że funkcja PRZESUNIĘCIE zwraca zakres o wielkości takiej jak zakres_odniesienia,
chyba że podamy inne wartości argumentów wysokość lub szerokość. Jeżeli wynik działania funkcji
obejmuje więcej niż jedną komórkę, należy pamiętać o użyciu klawiszy Ctrl+Shift+Enter, nawet
jeśli chcemy wykorzystać tylko pierwszą komórkę z wyniku. Ewentualnie można objąć wynik
funkcją L (N) lub T.

Gdy argumenty wiersze lub kolumny, a także wysokość lub szerokość są tablicami, funkcja
PRZESUNIĘCIE zwraca pseudozakres, który reprezentuje tablicę zakresów, ale nie może być bez-
pośrednio użyty w wyrażeniach. Można go użyć jako argumentu funkcji L (N) i T oraz SUMY.
CZĘŚCIOWE i AGREGUJ, funkcji zliczania i sumowania warunkowego, np. SUMA.JEŻELI, a także
wielu innych funkcji. Przekształcają one pseudozakres w tablicę, której można użyć w wyraże-
niach lub wyświetlić w arkuszu [B21].

Jeżeli chcemy wykorzystać funkcję PRZESUNIĘCIE w arkuszu z dużą ilością danych, należy pa-
miętać, że jest to funkcja ulotna (por. rozdział 5., podrozdział „Funkcje i działania ulotne”), a więc
jest obliczana przy każdym przeliczeniu arkusza. Jeśli zależy nam na szybkości działania arku-
sza, powinniśmy starać się, aby wywołanie funkcji PRZESUNIĘCIE było jak najprostsze, zaś jej ar-
gumenty, jeśli wymagają obliczenia, powinny być obliczane w osobnych komórkach, a tylko
wyniki przekazywane do funkcji. Dzięki temu unika się wielokrotnego niepotrzebnego oblicza-
nia argumentów.

Proste przykłady użycia funkcji PRZESUNIĘCIE
Przykład 6.3.
Wróćmy do przykładu 6.1, w którym wyodrębnialiśmy fragmenty obszaru lub tablicy przy użyciu
funkcji INDEKS. Jeśli chcemy w zdefiniowanym wcześniej obszarze TA pominąć pierwszy wiersz
i pierwszą kolumnę, możemy zastosować funkcję PRZESUNIĘCIE (tablica TA musi być zapisana
w arkuszu, aby miała adres):

= PRZESUNIĘCIE(TA;1;1;3;3)

W wyniku otrzymamy pełnowartościowe odwołanie do obszaru zawierającego takie liczby:

5 –3 7

8 10 –2

7 4 3

Przykład 6.4.
Funkcję PRZESUNIĘCIE można również wykorzystać do utworzenia odwołania do komórki o wyli-
czonym adresie. W tym przypadku należy użyć wyrażenia:

= PRZESUNIĘCIE($A$1; numer_wiersza – 1; numer_kolumny – 1)

Adres komórki odniesienia można poprzedzić kwalifikatorem arkusza.

Przykład 6.5.
Przykład 4.3. dotyczył połączenia dwóch obszarów danych: obsz1, obejmującego trzy kolumny,
oraz obsz2, złożonego z dwóch kolumn w jedną ciągłą tablicę pięciokolumnową. Zasygnalizo-
wałem tam możliwość użycia do tego funkcji PRZESUNIĘCIE. Teraz przyszła pora, aby tę zapo-
wiedź zrealizować.

Zadaniem funkcji PRZESUNIĘCIE jest poszerzenie obszaru obsz2 o trzy kolumny umieszczone
z przodu, przed właściwymi danymi. Należy w tym celu wywołać ją w taki sposób:

PRZESUNIĘCIE(obsz2;0;-3;;5)

Aby wyeliminować błędy, można wykorzystać — jak w przykładzie 4.3. — funkcję JEŻELI.BŁĄD.
Doprowadzi to nas do formuły tablicowej:

= JEŻELI.BŁĄD({1;1;1;0;0}*obsz1;0)+JEŻELI.BŁĄD({0;0;0;1;1}*PRZESUNIĘCIE(obsz2;0;-3; ;5);0)

Wzór można zastosować tylko do danych liczbowych (dane nieliczbowe zostaną zastąpione zera-
mi). Funkcja JEŻELI.BŁĄD musi być użyta dwukrotnie, osobno dla każdego obszaru. Formuła
po zatwierdzeniu tablicowym może być wykorzystana do dalszych obliczeń.

Do eliminacji błędów można wykorzystać również funkcje JEŻELI lub WYBIERZ. W tym przy-
padku nie musimy się ograniczać do danych liczbowych; mogą być one dowolnego typu.

= JEŻELI({1;1;1;0;0};obsz1;PRZESUNIĘCIE(obsz2;0;-3;;5))
= WYBIERZ({1;1;1;2;2};obsz1;PRZESUNIĘCIE(obsz2;0;-3;;5))

Otrzymane formuły są wyrażeniami TPA, więc mogą być wykorzystane do dalszych obliczeń.
Warunkiem powodzenia tych metod jest to, by obszar obsz2 nie był położony zbyt blisko lewej
krawędzi arkusza.

Metody tej nie można stosować do tablic, a jedynie do obszarów, które mają swój adres w arkuszu.

Odwracanie porządku elementów w zakresie
Inny przykład zastosowania funkcji PRZESUNIĘCIE to odwracanie porządku elementów w zakresie.
Wcześniej pokazałem, jak można ten problem rozwiązać za pomocą funkcji INDEKS.

Załóżmy, że mamy zakres wierszowy, któremu nadamy nazwę wiersz. Chcemy wygenerować za-
kres o identycznych wymiarach, w którym kolejność elementów będzie odwrócona. Zakres wy-
nikowy można umieścić w dowolnym miejscu arkusza. W pierwszej komórce tego zakresu umiesz-
czamy wywołanie funkcji:

= PRZESUNIĘCIE(wiersz;0;LICZBA.KOLUMN(wiersz)-NR.KOLUMNY(A1);1;1)

i kopiujemy w całym zakresie wyniku. Jest to zwykła formuła (nie tablicowa).

Analogiczną formułę można zastosować do zakresu kolumnowego o nazwie kolumna:

= PRZESUNIĘCIE(kolumna;ILE.WIERSZY(kolumna)-WIERSZ(A1);0;1;1)

Podobny efekt można osiągnąć za pomocą jednej formuły tablicowej:

= T(PRZESUNIĘCIE(kolumna; ILE.WIERSZY(kolumna)– WIERSZ(kolumna)+ KOMÓRKA("wiersz";
kolumna)-1;0))

Jako funkcji zewnętrznej należy użyć funkcji T, jeśli dane są tekstowe, a L lub N, gdy dane są liczbo-
we. Formuła w przypadku wiersza byłaby analogiczna. Tu wyrażenie z funkcją PRZESUNIĘCIE
należy do kategorii tzw. pseudozakresów, stąd potrzeba użycia dodatkowej funkcji, aby udo-
stępnić wartości formuły [B21].

Bez użycia funkcji
Początkowe elementy tablicy
Aby odwołać się do początkowych elementów tablicy (początkowych wierszy lub kolumn), nie
potrzeba używać żadnych funkcji. Wystarczy zaznaczyć miejsce na wynik, wpisać odwołanie do
tablicy i zatwierdzić przez Ctrl+Shift+Enter. W przypadku obszaru lub tablicy TA z przykładu
6.1., po zaznaczeniu obszaru złożonego z dwóch kolumn i trzech wierszy, wprowadzeniu wyra-
żenia = TA + 2 i zatwierdzeniu przez Ctrl+Shift+Enter, otrzymamy w wyniku obszar, zawiera-
jący następujące dane:

3 4

5 7

8 10

Jeżeli wynik operacji tablicowej umieszczamy w arkuszu, możemy wykorzystać dowolną liczbę
początkowych wierszy i kolumn faktycznego wyniku. Wszystko zależy od tego, ile miejsca na
wynik przygotujemy. Jednak, jeżeli wynik operacji tablicowej jest argumentem innej funkcji ta-
blicowej lub agregującej, to tablica w takim wyrażeniu zostanie uwzględniona w całości (o ile
jest to tablica pełnowartościowa).

Część wspólna zakresów
Elementy tablicy można również odczytać jako część wspólną wiersza lub zakresu wierszy i kolum-
ny lub zakresu kolumn. Jest to wskazane w przypadku, gdy kolumnom i wierszom tablicy na-
dano nazwy, jak w przykładzie pokazanym w rozdziale 2., na rysunku 2.4. Sposób ten można
wykorzystać do odczytania spójnego fragmentu tablicy, to znaczy fragmentu obejmującego są-
siadujące wiersze i kolumny.

W oparciu o dane widoczne na rysunku 2.4 można na przykład utworzyć odwołanie:

= T.top:T.wrz Etanol:Metanol

Obejmuje ono sześć komórek w trzech wierszach i dwu kolumnach. Po zaznaczeniu miejsca na
wynik zatwierdzamy formułę przez Ctrl+Shift+Enter.

Operator wyznaczania części wspólnej (spacja) może być również użyty do zakresów nieregu-
larnych. Jeśli wynikiem jest pojedyncza komórka, w arkuszu jest wyświetlana jej wartość i do
zatwierdzenia wystarczy sam klawisz Enter, jeśli zaś obszar, wyświetlana jest zawartość jego
komórek po zatwierdzeniu tablicowym w kilku komórkach. W przypadku, gdy wynikiem jest
zakres nieregularny, nie można go wyświetlić w arkuszu, możliwe jest natomiast użycie go jako
argumentu funkcji akceptującej zakresy nieregularne. Jeśli wynik ma być wyświetlony w jednej
komórce, nie jest potrzebne zatwierdzenie tablicowe.

Dynamiczne generowanie tablic i zakresów
o zadanych rozmiarach
Definicje tablic bazowych
Tablice jednowymiarowe (wierszowe lub kolumnowe) zawierające kolejne liczby naturalne od 1
do n są bardzo pomocne w obliczeniach dotyczących ciągów i szeregów, jak również w wielu in-
nych, które wykorzystują te tablice w sposób pośredni. Będę je nazywał dalej tablicami bazo-
wymi (nie jest to termin powszechnie używany, wprowadziłem go na potrzeby tej publikacji).

Wygenerowanie takich tablic wymaga posłużenia się pewną sztuczką, gdyż nie ma funkcji słu-
żącej specjalnie do tego celu. Można mianowicie wykorzystać jedną z funkcji, których pierwotne
przeznaczenie było inne. Są to WIERSZ(przedział) lub NR.KOLUMNY(przedział). Omówiłem je
szczegółowo w rozdziale 3., w podrozdziale „Numery wiersza i kolumny w arkuszu”. Funkcje te,
użyte w formule tablicowej, zwracają listę numerów wierszy lub kolumn podanych jako argument.
Wynik ma formę tablicy jednowymiarowej, niezależnie od wymiarów przedziału. W przypad-
ku funkcji WIERSZ jest to wektor kolumnowy, a w przypadku NR.KOLUMNY — wektor wierszowy.
Z użyciem tych funkcji wiąże się jednak pewna niedogodność — jeżeli usunie się wiersze bądź
kolumny z przedziału użytego jako argument lub poprzedzające go (lub doda się w tych miejscach
nowe), przedział zostanie zmodyfikowany, podobnie jak modyfikowane są w takim przypadku
wszystkie adresy w arkuszu. Aby zabezpieczyć się przed tą niedogodnością, a ponadto uzyskać moż-
liwość konstrukcji przedziału z segmentów, należy użyć jeszcze funkcji ADR.POŚR(tekst_adresu),
podając tekst_adresu jako stałą tekstową w cudzysłowie, wyrażenie tekstowe lub odwołanie do
komórki z tekstem adresu. Opis składni i sposób użycia tej funkcji podałem w rozdziale 3.,
w podrozdziale „Funkcja ADR.POŚR”.

W ten sposób, używając na przykład wyrażenia:
= WIERSZ(ADR.POŚR("1:8")) lub = WIERSZ(ADR.POŚR("A1:A8"))

uzyskamy tablicę ={1\2\3\4\5\6\7\8} (od wersji 2010 ={1;2;3;4;5;6;7;8}). Aby „uzmiennić”
górną granicę zakresu, należy zamiast stałej 8 wpisać odwołanie do komórki zawierającej tę
liczbę. Trzeba pamiętać, że argument jest wyrażeniem tekstowym i trzeba użyć operatora złączenia
tekstów &. Wartość w komórce może być wartością numeryczną, Excel sam skonwertuje ją do
postaci tekstowej. Jeśli założymy, że wartość 8 umieściliśmy w komórce A5, zmodyfikowane wy-
rażenie może przyjąć postać:

= WIERSZ(ADR.POŚR("1:"&$A$5))

a jeżeli nadaliśmy komórce A5 nazwę n, jedną z postaci:

= WIERSZ(ADR.POŚR("1:"&n))
= WIERSZ(ADR.POŚR("A1:A"&n))

Standardowo przy nadawaniu nazwy komórce jej adres jest zapamiętywany jako adres absolut-
ny. Można jednak ręcznie — za pomocą Menedżera nazw — zmienić to odwołanie na względne
lub mieszane.

Formuła jest dynamiczna w tym sensie, że zmiana wartości n w arkuszu automatycznie modyfi-
kuje rozmiar tablicy, przy czym jest on ograniczony tylko liczbą wierszy w arkuszu (1 ≤ n ≤
65 536 w Excelu do wersji 2003 oraz 1 ≤ n ≤ 1 048 576 od wersji 2007).

Lista numerów wierszy generowana przez funkcję WIERSZ zawsze będzie uporządkowana
rosnąco; nie pomoże przestawienie kolejności numerów wierszy typu = WIERSZ(ADR.
POŚR("8:1")). Jeśli potrzebna jest odwrócona kolejność, należy użyć wyrażenia w ro-
dzaju = 9 – WIERSZ(ADR.POŚR("1:8")).

Teraz sprawa kluczowa! Tego wyrażenia nie należy wpisywać do arkusza (a przynajmniej nie
wyłącznie do arkusza), bo wtedy długość tablicy będzie ograniczona wielkością zaznaczonego
zakresu. Zamiast tego, trzeba je zapamiętać pod wybraną nazwą, na przykład wn, posługując się
poleceniem Wstaw/Nazwa/Definiuj (od wersji 2007 Formuły/Definiuj nazwę), a następnie w polu
Nazwy w skoroszycie: wpisać wybraną nazwę, a w polu Odwołuje się do: podane powyżej wyra-
żenie, po czym zatwierdzić to przyciskiem OK. Jest to formuła tablicowa, jednak przy definio-
waniu nazwy nie trzeba używać kombinacji Ctrl+Shift+Enter. Ponadto wyrażenia z nazwą wn są
krótsze i bardziej czytelne od tych zawierających konstrukcję WIERSZ(ADR.POŚR("1:"&n)).

Jeśli wyrażenie typu wn chcemy wyświetlić w arkuszu, należy pamiętać, że pełną tablicę uzyskamy po
jej wprowadzeniu do dokładnie n komórek. Gdy będzie ich mniej, końcowe liczby nie zostaną
wyświetlone; jeśli będzie ich więcej, w „nadmiarowych” pojawią się błędy #N/D!. Aby uniknąć
wyświetlania tych błędów, można rozbudować formułę do postaci:

= JEŻELI(WIERSZ()-WIERSZ(adres_pocz)<n; WIERSZ(ADR.POŚR("1:"&n)); "")

Adres_pocz oznacza adres pierwszej komórki, w której wpisujemy formułę. Jeśli chcemy całość
później kopiować, powinien to być adres względny. Aby formalnie uniezależnić formułę od kon-
kretnego adresu początkowego, można posłużyć się adresem pośrednim w stylu W1K1:

= JEŻELI(WIERSZ()-WIERSZ(ADR.POŚR("WK";))<n; WIERSZ(ADR.POŚR("1:"&n)); "")

Czasem wygodniej jest, gdy ciąg liczb zaczyna się od zera; nie można oczywiście wpisać 0 jako
argumentu funkcji ADR.POŚR, bo nie ma zerowego wiersza, ale gdy funkcja WIERSZ już „zrobi swoje”,
można odjąć jedynkę:

wn0 : = WIERSZ(ADR.POŚR("1:"&n+1))-1

Alternatywnymi definicjami wn mogą być formuły:

= WIERSZ($A$1:INDEKS($A:$A;n))
= WIERSZ(PRZESUNIĘCIE($A$1;;;n))

Ich wadą jest podatność na wstawianie lub usuwanie pierwszego wiersza w arkuszu, zaletą —
brak funkcji ADR.POŚR, która czasem jest niepożądana.

Jeżeli przy definiowaniu n wykorzystano adres absolutny, to użycie w dowolnej komórce nazwy
wn będzie oznaczać taką samą kolumnę (pionową tablicę) liczb od 1 do n. Jeżeli natomiast przy
definiowaniu n użyto adresu względnego, a komórką aktywną w chwili definiowania była ko-
mórka sąsiednia z prawej strony n, to użycie w dowolnej komórce nazwy wn będzie oznaczać

kolumnę (pionową tablicę) liczb od 1 do n, przy czym n będzie liczbą pobraną z komórki sąsia-
dującej z lewej strony z komórką, w której użyto nazwy wn. Jeżeli wynik będzie wyświetlany tablico-
wo w kilku komórkach, bierze się pod uwagę pierwszą z nich. Więcej informacji na ten temat
podano w rozdziale 7., w podrozdziale „Zasady interpretacji nazw i adresów w wyrażeniach ta-
blicowych”.

Tablica wn zdefiniowana w powyższy sposób jest więc tablicą kolumnową. W razie potrzeby do
zdefiniowania analogicznej tablicy wierszowej można użyć funkcji TRANSPONUJ(wn). W zasadzie
można też użyć funkcji NR.KOLUMNY, ale jako argument należałoby podać oznaczenia literowe
kolumn, co jest dużo mniej wygodne od numerów wierszy. Można też użyć dodatkowo funkcji
PRZESUNIĘCIE:

= NR.KOLUMNY(PRZESUNIĘCIE($A$1;0;0;1;n))

albo w wersji uproszczonej:

= NR.KOLUMNY(PRZESUNIĘCIE($A$1;;;;n))

Zamiast funkcji PRZESUNIĘCIE można użyć również funkcji INDEKS w taki sposób:

= NR.KOLUMNY($A$1:INDEKS($1:$1;n))

Należy jednak pamiętać, że ostatnie wzory są podatne na modyfikacje przy wstawianiu lub usuwa-
niu wierszy lub kolumn.

Można również użyć zapisu:

= NR.KOLUMNY(ADR.POŚR("A1:"&ADRES(1;n)))

Ze względu na użycie funkcji ADR.POŚR ten zapis jest odporny na modyfikacje struktury arkusza.

Tablice bazowe można również zdefiniować jako funkcje UDF w VBA [C4].

Wykorzystanie tablic bazowych do definicji innych tablic
Dysponując tablicą bazową zawierającą liczby od 1 do n, możemy definiować inne potrzebne
tablice jedno- i dwuwymiarowe. W oparciu o definicję wektora wn

wn : = WIERSZ(ADR.POŚR("1:"&n))

można zdefiniować na przykład macierz jednostkową E o rozmiarze n×n za pomocą formuł
podanych w rozdziale 4., w punkcie „Macierz jednostkowa”.

Wyrażenia te najlepiej zdefiniować, posługując się formułami nazwanymi za pomocą polecenia
Wstaw/Nazwa/Definiuj (od wersji 2007 Formuły/Definiuj nazwę). n musi być nazwą stałej zde-
finiowanej lub komórki zawierającej rozmiar macierzy. Następnie definiujemy wektor kolum-
nowy wn i wreszcie macierz E.

W Excelu 2013+ zadanie to można wykonać znacznie prościej, dzięki nowej funkcji predefiniowa-
nej. Wystarczy napisać E : = MACIERZ.JEDNOSTKOWA(n)

Po wykonaniu zwykłego mnożenia dowolnej tablicy wierszowej lub kolumnowej o długości n przez
macierz jednostkową, otrzymamy macierz diagonalną z elementami diagonalnymi równymi
elementom użytej tablicy.

Posługując się zdefiniowaną wcześniej tablicą wn, można łatwo zdefiniować inne często po-
trzebne tablice, na przykład:
 tablica złożona z n zer: = wn–wn albo = wn*0
 tablica złożona z n jedynek: = wn–wn+1 albo = wn*0+1, albo = wn^0, albo = 1^wn
 tablica złożona z n jedynek i minus jedynek na przemian: =-1^wn albo =-1^(wn+1),

albo =-(-1^wn)

Pierwszy wariant ostatniego wzoru zwraca ciąg rozpoczynający się od –1, dwa następne star-
tują od 1.

W tym przypadku problemem nie jest uzyskanie konkretnej wartości, lecz jej powtórzenie
w tablicy ściśle określoną liczbę razy.

Jeśli początkowa liczba w tablicy bazowej wn ma być inna niż 1, można „uzmiennić” obie grani-
ce lub użyć wyrażeń arytmetycznych typu wn + 5 bądź wn – n/2.

Jeżeli w definicji tablicy wn użyto zmiennej n, która ma aktualnie wartość 5, to wyrażenie:

= a+(wn–1)*(b–a)/(n–1)

definiuje kolumnę pięciu wartości rozłożonych równomiernie pomiędzy a i b. Analogicznie,
posługując się definicją wn0 i wartością n = 4, można podobny efekt uzyskać po obliczeniu:

= a+wn0*(b–a)/n

Temu wyrażeniu można nadać nową nazwę, na przykład x. x pozostanie wyrażeniem dynamicznym
tak długo, jak długo jest używane jako nazwa. Każda zmiana wartości n spowoduje zmianę
rozmiarów i zawartości x. Jeżeli jednak wprowadzimy x do arkusza, wpisując do komórki lub
zakresu komórek = x i zatwierdzając przez Enter lub Ctrl+Shift+Enter, ta komórka czy zakres nie
będzie reprezentować już całej tablicy x, a tylko to, co jest wyświetlone w arkuszu, czyli pierw-
szą lub kilka pierwszych komórek tablicy x. Rozmiar tego obszaru pozostaje statyczny, czyli nie
zmienia się przy zmianie n.

Każda formuła tablicowa w danej chwili musi dawać w wyniku tablicę o określonej wielkości
i kształcie, zależnie od składowych formuły i zasad wykonywania działań. Z formułą, która repre-
zentuje tablicę dynamiczną, nie jest jednak związany żaden adres, w związku z tym ani nazwa
tablicy dynamicznej, ani nazwa takiej formuły nie może być argumentem funkcji, która wyma-
ga podania adresu, jak choćby funkcja PRZESUNIĘCIE.

Plan eksperymentu czynnikowego
Przykładem zastosowania tablic dynamicznych może być wygenerowanie planu eksperymentu
czynnikowego typu 2k. Jest to jeden z podstawowych planów stosowanych do badania zależno-
ści między różnymi wielkościami. Plan jest tablicą, która ma tyle kolumn, ile jest zmiennych

niezależnych, oraz jedną dodatkową kolumnę wyrazu wolnego. Liczba wierszy odpowiada liczbie
doświadczeń, które należy wykonać, równej 2k. Wartości czynników są kodowane liczbami 1 lub –1,
przy czym chodzi o to, by zrealizować wszystkie możliwe kombinacje (w znaczeniu potocznym,
nie kombinatorycznym) tych liczb. Wyjątkiem jest pomocnicza kolumna wyrazu wolnego,
umieszczana zwykle jako pierwsza, która zawiera same jedynki.

Liczbę czynników wstawiono do komórki C3, której nadano nazwę lk. Następnie zdefiniowano dwa
wektory — jeden wierszowy o nazwie wlk, zawierający liczby od 0 do lk, a drugi kolumnowy o
nazwie nlk, złożony z liczb od 1 do 2lk. Definicje tych wektorów mają postać:

wlk : = TRANSPONUJ(WIERSZ(ADR.POŚR("1:"&lk+1))–1

albo wlk : = NR.KOLUMNY(ADR.POŚR("A1:"&ADRES(1;lk+1)))-1

nlk : = WIERSZ(ADR.POŚR("1:"&2^lk))

Wreszcie, w oparciu o wektory wlk i nlk, zdefiniowano tablicę plan, obejmującą wlk kolumn i nlk
wierszy, zawierającą niezbędne kombinacje dodatnich i ujemnych jedynek. Tablicę opisano
wzorem:

plan : = –1^ZAOKR.DO.CAŁK((nlk–1)/2^(lk–wlk))

Funkcja ZAOKR.DO.CAŁK zwraca część całkowitą wyniku dzielenia nlk–1 przez odpowiednią potęgę
dwójki. Jeżeli wynik funkcji jest liczbą parzystą, formuła przyjmuje wartość 1, a jeśli liczbą nie-
parzystą — zwracana jest wartość –1. Na rysunku pokazano całą tablicę plan oraz wygenerowa-
ne na jej podstawie dodatkowe kolumny iloczynów wybranych czynników.

Taki plan wygenerowany dla k = 4 pokazano na rysunku 6.1.

Selektywne sumowanie elementów tablic dwuwymiarowych
Użycie funkcji SUMA wprost do elementów tablicy dwuwymiarowej prowadzi do obliczenia su-
my ogólnej wszystkich elementów tablicy. W obliczeniach często potrzebne są sumy wierszy
lub kolumn. W takim przypadku należy wykorzystać dodatkowe funkcje do selekcji danych.
Najprostszym rozwiązaniem jest użycie funkcji INDEKS do wyodrębnienia wiersza lub kolumny
do sumowania. Jeżeli chcemy obliczyć sumę określonego wiersza lub kolumny, zastosujemy
formułę:

= SUMA(INDEKS(tabl; nr_wiersza;))

lub

= SUMA(INDEKS(tabl;;nr_kolumny))

Są to formuły zwykłe, każda zwraca jedną wartość — odpowiednio dla wiersza lub kolumny.

Jeśli chcemy uzyskać wynik w postaci tablicy (kolumny z podsumowaniem wierszy lub wiersza
z podsumowaniem kolumn), funkcja INDEKS musiałaby być użyta w sposób niestandardowy (z ar-
gumentem indeksowym w postaci tablicy) [B20].

RYSUNEK 6.1. Plan całkowitego eksperymentu czynnikowego dla czterech czynników, wygenerowany
za pomocą formuły tablicowej

Na szczęście istnieje też sposób alternatywny — z wykorzystaniem funkcji MACIERZ.ILOCZYN.
Mieści się on w standardzie użycia tej funkcji, ale jest nieco mniej intuicyjny i nadaje się tylko do
sumowania. Sposób jej działania opisałem w rozdziale 4., w punkcie „Mnożenie macierzowe”.

Użycie funkcji MACIERZ.ILOCZYN polega na przemnożeniu macierzowym tablicy tabl przez ko-
lumnę lub wiersz, złożone z samych jedynek. Wynik otrzymujemy w formie kolumny lub wiersza.

Należy zastosować jedną z formuł tablicowych:

= MACIERZ.ILOCZYN(tabl; TRANSPONUJ(NR.KOLUMNY(tabl))^0)

w odniesieniu do sum wierszy lub

= MACIERZ.ILOCZYN(TRANSPONUJ(WIERSZ(tabl)^0); tabl)

w odniesieniu do sum w kolumnach.

Formuły z funkcją MACIERZ.ILOCZYN są pełnowartościowe, to znaczy mogą stać się elementem więk-
szego wyrażenia.

Łączenie tablic z wykorzystaniem funkcji MACIERZ.ILOCZYN
Już kilkakrotnie zajmowaliśmy się zagadnieniem łączenia obszarów i tablic w jedną spójną tablicę.
Wiemy już, że stosunkowo łatwo można połączyć zakresy lub tablice jednowymiarowe (wiersze
lub kolumny), a w przypadku obszarów można za pomocą funkcji PRZESUNIĘCIE połączyć również
struktury dwuwymiarowe. Najwięcej problemów przysparzają tablice dwuwymiarowe, które
nie są umieszczone bezpośrednio w arkuszu, lecz są wynikiem obliczeń. Jak pokazałem w przy-
kładzie 6.2, można je połączyć za pomocą funkcji INDEKS po dekompozycji na pojedyncze wiersze
lub kolumny. W przypadku większych tablic sposób ten jednak nie jest zbyt wygodny. Teraz,
po zapoznaniu się z możliwościami wykorzystania tablic bazowych, przyszedł czas na omówie-
nie jeszcze jednej metody łączenia tablic z użyciem funkcji MACIERZ.ILOCZYN. Dzięki temu, że
jest to funkcja tablicowa, nie podlega ograniczeniom wynikającym z zasady superpozycji i od-
powiednio zaprogramowana, może wykonać nasze zadanie. Metodę tę można stosować zarów-
no do tablic, jak i do obszarów, jest jednak konieczne, aby dane były wyłącznie liczbowe. W za-
kresie danych nie może być również pustych komórek (funkcja MACIERZ.ILOCZYN jest bardzo
wrażliwa na błędne dane).

Omówię teraz pokrótce ideę rozwiązania.

Najpierw ustalamy rozmiary tablicy wynikowej i upewniamy się, czy łączone tablice są dopaso-
wane wymiarami. Tablice łączone w strukturę poziomą (kolumna obok kolumny) muszą mieć
taką samą liczbę wierszy, zaś łączone w strukturę pionową (wiersz pod wierszem) muszą mieć taką
samą liczbę kolumn.

Przy tworzeniu struktury poziomej obliczamy sumaryczną liczbę kolumn w połączonej tablicy i de-
finiujemy poziomą tablicę bazową z numerami kolumn. Następnie do każdej tablicy składowej
definiujemy pionowe tablice bazowe z liczbami od 1 do liczby kolumn w każdej tablicy składowej.
Przy tworzeniu struktury pionowej postępujemy analogicznie, tylko z zamianą wierszy na ko-
lumny i odwrotnie.

Generujemy odpowiednie tablice zero-jedynkowe, które pomnożone macierzowo przez tablice
składowe dadzą w wyniku tablice o rozmiarach tablicy wynikowej, odwzorowujące tablice składowe
uzupełnione zerami.

Sumujemy tablice cząstkowe, uzyskując tablicę wynikową.

Przykład 6.6.
Rozpatrzymy połączenie poziome trzech tablic: TA, TB i TC. Tablica TA ma a kolumn, TB — b kolumn,
a TC — c. kolumn. Liczba wierszy nie ma znaczenia, musi być jedynie taka sama w każdej tabli-
cy. Łączna liczba kolumn tablicy wynikowej jest równa s = a + b + c.. Definiujemy poziomą
tablicę bazową o długości s:

ks : = TRANSPONUJ(WIERSZ(ADR.POŚR("1:"&s)))

oraz pionowe tablice bazowe o długościach odpowiadających liczbie kolumn każdej z tablic
składowych:

wa : = WIERSZ(ADR.POŚR("1:"&a))
wb : = WIERSZ(ADR.POŚR("1:"&b))
wc : = WIERSZ(ADR.POŚR("1:"&c.))

Tworzymy cząstkowe tablice wynikowe, odpowiadające tablicom składowym uzupełnionym zera-
mi, które następnie dodajemy:

= MACIERZ.ILOCZYN(TA; --(ks=wa)) + MACIERZ.ILOCZYN(TB; --(ks=wb+a)) +
MACIERZ.ILOCZYN(TC; --(ks=wc+b+a))

Wynik jest tablicą TPW, może być więc używany do dalszych obliczeń i nie wymaga zatwier-
dzenia tablicowego, jeśli tylko wynik końcowy jest pojedynczą liczbą. Można przykładowo użyć
takiej tablicy jako drugiego argumentu funkcji WYSZUKAJ.PIONOWO. Jeśli rozmiary tablic składo-
wych są stałe i niezbyt wielkie, wygodniejsze może być użycie odpowiednich stałych tablico-
wych zamiast formuł wa, wb i wc. Procedura łączenia tablic została zilustrowana na rysunku 6.2.
Tu użyto do połączenia obszarów, aby przekaz był czytelny, ale obszary mogą być używane na rów-
ni z tablicami.

RYSUNEK 6.2. Łączenie tablic z wykorzystaniem funkcji MACIERZ.ILOCZYN

Zakresy dynamiczne
W poprzednich punktach przedstawiłem sposób definiowania tablic o zmiennych rozmiarach.
Taka konstrukcja jest wystarczająca do wielu obliczeń, ale w niektórych przypadkach konieczne
jest odwołanie do zakresu komórek w arkuszu. Należy zatem umieć definiować zakres również
w sposób dynamiczny. Można sobie wyobrazić kilka możliwych sposobów realizacji tego zagadnie-
nia. Przedstawię je kolejno.

Różne typy adresu początku i końca
Jak już wcześniej wspomniałem, przy definiowaniu zakresu można na przykład jego początek
opisać odwołaniem absolutnym, a koniec — względnym lub mieszanym. W takim przypadku
rozmiary zakresu będą zależeć od tego, która komórka była aktywna w momencie definiowania
i w której użyto odwołania. Załóżmy na przykład, że komórką aktywną jest C1 i za pomocą Mene-
dżera nazw tworzymy definicję:

obszar : = $B$1:B$1

Tak zdefiniowany obszar oznacza zakres komórek od B1 do komórki w pierwszym wierszu ar-
kusza i kolumnie sąsiadującej z lewej strony z kolumną komórki, w której użyto nazwy w formule.
Jeżeli w komórce E3 umieścimy formułę

= ŚREDNIA(obszar)

to obliczona zostanie średnia wartości z zakresu B1:D1. Jeśli skopiuje się formułę w prawo do
komórek w kolejnych kolumnach, uśredniany zakres będzie coraz większy.

Ten sposób definiowania zakresów dynamicznych jest najprostszy, ale zarazem najmniej uni-
wersalny. Przy kopiowaniu danych w inne miejsce na ogół konieczna jest ręczna korekta formuł.

Stały początek obszaru, długość określona przez położenie formuły
Ogólną koncepcję opisaną w poprzednim podpunkcie można zrealizować również formułą,
wykorzystując jako początek zakresu komórkę z charakterystyczną zawartością lub pustą, a ko-
niec uzależniając od położenia komórki z formułą. Wiele formuł dynamicznych konstruowa-
nych jest w taki sposób, że odnoszą się do zakresu komórek położonych nad lub przed komórką (po
jej lewej stronie), w której znajduje się formuła. Przykładowo, wyobraźmy sobie, że umieszcza-
my formuły w kolumnie E, przy czym pozostawiamy nad nimi pustą komórkę. Zakres dynamiczny
ma obejmować obszar od pustej komórki do znajdującej się bezpośrednio nad naszą formułą.
Załóżmy, że pusta komórka ma adres E5, a pierwszą formułę wprowadzamy do E6. W pierwszej
komórce zakres będzie obejmował tylko komórkę E5, a w miarę kopiowania w dół będzie się
wydłużał. Początek obszaru określimy jako ostatnią pustą komórkę w kolumnie E przed komórką
bieżącą. Zrealizujemy to za pomocą formuły, której można nadać nazwę i wykorzystywać jako
element złożonych obliczeń.

zw : = INDEKS(E$1:E5; MAX(JEŻELI(CZY.PUSTA(E$1:E5);WIERSZ(E$1:E5)))):E5

Przy wprowadzaniu tej definicji komórką aktywną jest E6. Formuła jest tablicowa, ale jeśli nadamy
jej nazwę, nie będzie to miało praktycznego znaczenia. Zakładam, że użytkownik nie będzie
wstawiał ani usuwał pierwszego wiersza. Dzięki użyciu względnego adresu kolumny, po zapisa-
niu definicja dopasowuje się do komórki, w której jest używana.

Stały początek obszaru, długość określona przez liczbę komórek
Wyobraźmy sobie, że tablicę bazową wn, zdefiniowaną jako:

wn : = WIERSZ(ADR.POŚR("1:"&n))

wprowadzimy do wybranej kolumny arkusza, zaczynając na przykład od komórki C6, której może-
my nadać nazwę pocz. Zakres, do którego wprowadzono formułę, powinien być zaplanowany
z zapasem, gdyż określa on maksymalną wartość n, jakiej można w praktyce użyć. Na ogół nie
ma sensu odwoływać się do takiego zakresu w całości. Zawiera on tylko n wartości, pozostałe
komórki wypełnione są kodem braku danych #N/D!. Chcemy tak zdefiniować nasz zakres, aby
obejmował tylko n początkowych komórek, zawierających liczby od 1 do n.

Mamy w Excelu trzy funkcje, które można wykorzystać w tym celu. Są to INDEKS, ADR.POŚR
i PRZESUNIĘCIE. Zgodnie ze składnią tych funkcji możemy zapisać formuły:

= pocz:INDEKS(zakres; n)
= pocz:ADR.POŚR(ADRES(WIERSZ(pocz)+n-1; NR.KOLUMNY(pocz))
= pocz:ADR.POŚR("W" & WIERSZ(pocz)+n-1 & "K" & NR.KOLUMNY(pocz);)
= PRZESUNIĘCIE(pocz; 0; 0; n)

Drugą formułę można uprościć, jeśli uwzględni się konkretny adres początku zakresu:

= ADR.POŚR("C6:C" & n+5)

Jeśli chcemy, aby zakres rozpoczynał się nie od konkretnej komórki, ale na przykład bezpośrednio
pod komórką, w której wpiszemy odwołanie do niego, można posłużyć się formułą z adresem
pośrednim zapisanym w stylu W1K1:

= ADR.POŚR("W[1]K:W[" & n & "]K";)

Każda z powyższych formuł zwraca zakres obejmujący n komórek. Formule takiej można nadać
nazwę i używać jej wszędzie tam, gdzie wymagane jest użycie odwołania, czyli jako argument
funkcji odwołaniowej lub przy definiowaniu źródła listy rozwijanej. Istotne przy tym jest, żeby
zakres w arkuszu był faktycznie wypełniony odpowiednimi danymi. Gdyby jednak taka ko-
lumna psuła estetykę arkusza, można ją ukryć.

Formuły tablicowe uwzględniające faktyczne rozmiary
obszaru formuły
Zazwyczaj, gdy zapisujemy formułę tablicową, z góry przewidujemy rozmiary obszaru wyni-
kowego i rezerwujemy w arkuszu tyle miejsca, ile potrzeba do wyświetlenia wszystkich wyni-
ków. Gdy zaznaczymy zbyt duży obszar, w tych dodatkowych komórkach pojawią się kody błę-
dów (najczęściej #N/D!). Zwykle staramy się unikać takich sytuacji ze względów estetycznych.
Możemy wykorzystać różne sposoby maskowania błędów, opisane w rozdziale 8., w punkcie „Ma-
skowanie wartości błędów”. Problemy stwarzają szczególnie formuły dynamiczne, bo w ich przy-
padku rozmiary obszaru wynikowego są zmienne, a jeśli chcemy pokazać wyniki w arkuszu, to
musimy użyć do tego obszaru statycznego.

Jeśli weźmiemy pod uwagę typową formułę tablicy bazowej:

wn : = WIERSZ(ADR.POŚR("1:"&n))

w której n reprezentuje wartość liczbową, to w formułach złożonych formuła ta oznacza tablicę
dokładnie n-elementową i jeśli będzie argumentem funkcji zwykłej, w arkuszu wymagać będzie
dokładnie n komórek. Jak zmodyfikować tę formułę, aby w przypadku zaznaczenia większej

liczby komórek, w tych nadmiarowych wyświetlane były puste teksty zamiast błędów? Ta zmodyfi-
kowana formuła musi umieć odczytać nie tylko wartość n, ale również rozmiary obszaru, do którego
została wprowadzona. Takich informacji dostarczyć mogą funkcje WIERSZ() lub NR.KOLUMNY(),
użyte bez argumentów. Wyrażenie:

lnr : = WIERSZ()-WIERSZ(ap)+1

gdzie ap jest adresem komórki, w której rozpoczyna się zakres formuły, zwraca listę kolejnych
liczb całkowitych od 1 do wypełnienia całego zakresu. Zamiast podawać adres ap bezpośrednio,
można posłużyć się formułą funkcji ADR.POŚR z argumentem w stylu W1K1:

lnr : = WIERSZ()-WIERSZ(ADR.POŚR("WK";))+1

Formuły tego rodzaju mogą być wygodnie używane do wyświetlania w arkuszu tylu wyników, ile
chcemy zaznaczyć wierszy. Ale tego typu formuły nie nadają się do wykorzystania jako argu-
menty funkcji agregujących. Gdybyśmy chcieli obliczyć sumę numerów zaznaczonych wierszy
od 1 do 7 (przykładowo), to formułę

SUMA(lnr)

musielibyśmy wprowadzić do siedmiu komórek i otrzymalibyśmy siedem identycznych wyników.

Jeśli chcielibyśmy ograniczyć długość ciągu liczbowego do n wartości, powinniśmy wykorzystać
funkcję JEŻELI:

= JEŻELI(lnr <= n; lnr; "")

Aby uzyskać formułę w miarę uniwersalną, którą można by wykorzystywać zarówno w funk-
cjach zwykłych, jak i agregujących, należy połączyć oba podejścia w jednej formule:

= JEŻELI(lnr <= n; wn; "")

Gdy chcemy użyć tej konstrukcji z funkcją zwykłą, musimy ją wprowadzić do wnętrza funkcji
JEŻELI, na przykład:

= JEŻELI(lnr <= n; LN(wn); "")

Funkcję agregującą można umieścić podobnie lub na zewnątrz — jeśli wyrażenie będzie za-
twierdzane w jednej komórce, forma:

= JEŻELI(lnr <= n; SUMA(wn); "")

nie wymaga zatwierdzenia tablicowego (o ile na wn nie są wykonywane żadne operacje), nato-
miast forma:

= SUMA(JEŻELI(lnr <= n; wn; ""))

wymaga takiego zatwierdzenia.

Poniżej przytaczam inne warianty formuł wykorzystujących funkcje WIERSZ() lub NR.KOLUMNY().
Przykładowo, obliczając wartości wyrażeń:

lw : = MAX(WIERSZ())-MIN(WIERSZ())+1

lub

SUMA(WIERSZ())/ŚREDNIA(WIERSZ())

możemy określić liczbę wierszy, które zajmuje formuła tablicowa.

Analogicznie można obliczyć liczbę kolumn:

lk : = MAX(NR.KOLUMNY())-MIN(NR.KOLUMNY())+1

Proszę zauważyć, że nie można tu wykorzystać funkcji ILE.WIERSZY ani LICZBA.KOLUMN, gdyż
one wymagają, aby adres zakresu był jawnie określony.

Można teraz wykorzystać formuły nazwane lw i lk, na przykład do utworzenia kopii zakresu
komórek, począwszy od znanej komórki początkowej (na przykład A3), obejmującej tyle ko-
mórek, ile zaznaczymy. Można to zrobić kilkoma sposobami, stosując jedną z funkcji PRZESUNIĘCIE,
INDEKS lub ADR.POŚR.

= PRZESUNIĘCIE($A$3;;; lw; lk)
= $A$3:PRZESUNIĘCIE($A$3; lw-1; lk-1)
= $A$3:INDEKS($A$3:$AA$100; lw; lk)
= ADR.POŚR("A3:"&INDEKS(ADRES(WIERSZ($A$3)+lw-1; NR.KOLUMNY($A$3)+lk-1);1))

W przypadku formuły z funkcją INDEKS należy podać w argumencie zakres dostatecznie duży,
aby objął on planowaną kopię. Występująca w przypadku formuły z funkcją ADR.POŚR potrzeba
użycia dodatkowo funkcji INDEKS została wyjaśniona w rozdziale 7., w przykładzie 7.21.

Podane cztery formuły są pełnowartościowe i mogą być wykorzystane jako argumenty innych
funkcji. Jeśli jednak użyjemy ich jako argumentu funkcji agregującej, będziemy musieli zatwierdzić
formułę w całym obszarze określającym wielkość kopii, otrzymamy zatem wynik powtórzony
wielokrotnie. Można tego uniknąć, dopisując do formuły warunek wyświetlenia wyniku tylko
w pierwszej komórce:

= JEŻELI((WIERSZ()=MIN(WIERSZ()))*(NR.KOLUMNY()=MIN(NR.KOLUMNY()));SUMA(kopia);"")

W powyższym wzorze kopia oznacza jedną z czterech formuł podanych powyżej, a SUMA została
użyta jako przykład funkcji agregującej.

Inne typy formuł uwzględniających rozmiary obszaru, w którym zostały zatwierdzone, wyko-
rzystują struktury pseudozakresów [B21].

aktywny, 554
alternatywa kryteriów, 403
analiza

ciągów, 328
danych, 449
zawartości arkusza, 292

Analysis ToolPak, 141
aparat fotograficzny, 275
apostrof, 91
aproksymacja funkcji, 555
argumenty

funkcji, 77, 214, 555
grupowe, 123, 125, 130
niedopasowane, 111
opcjonalne, 227
puste, 227, 228
w postaci tablic, 126

arkusz kalkulacyjny, 23, 556
analiza zawartości, 292
odwołania wielopoziomowe, 142
przeliczanie, 179
rodzaje informacji, 31
struktura, 31
wyświetlanie danych, 147
zmiana wyglądu, 182

ASCII, 52, 75
autofiltr, 399
automatyczne

formatowanie, 501
tworzenie hiperłączy, 500

autouzupełnianie formuł, 72

B
baza danych, 389, 475

właściwości pierwiastków chemicznych, 393
z tabelą danych, 448

bieżący, 554
błąd

#ADR! , 97, 280
#ARG! , 108, 250, 280
#DZIEL/0!, 280
#LICZBA! , 80, 280
#N/D! , 111, 280
#NAZWA?, 280
#ZERO!, 280

błędy
identyfikacja przyczyn, 294
kody, 278
kontrola, 278
obliczeń, 280
sygnalizacja, 278

C
ciąg, 556

cyfr, 437
cieniowanie wierszy, 382
czas, 36, 352, 354

systemowy, 356
czcionki, 182
część wspólna zakresów, 55
czujki, 295

D
dane

grupowe, 105, 247, 248
liczbowe uporządkowane, 316, 321
nieliczbowe, 327
rozproszone, 332
w tablicy, 461
w zakresie, 459
zewnętrzne, 46, 48

data, 36, 352, 355
definiowanie

formatów warunkowych, 362
kwerendy, 477
tablic bazowych, 197

dereferencja, 234, 557
diagnozowanie błędów obliczeń, 278
dodatek Analysis ToolPak, 141
dwukropek, 106
dynamiczne generowanie tablic, 197
dysk sieciowy OneDrive, 27
działania arytmetyczne, 110

E
edycja, 153

formuł tablicowych, 176
elementy

składowe tabeli, 134
tablicy, 105

F
filtr

nazw, 62
zaawansowany, 401

alternatywa kryteriów, 403
koniunkcja kryteriów, 403
kryteria z nagłówkami kolumn, 402

filtrowanie, 390, 392
formanty pola kombi, 276
format, 559

Czas, 36
Data, 36
Liczbowe, 33, 344, 348
Naukowe, 34
Ogólne, 33, 36, 344
Procentowe, 35
Tekstowe, 37, 348
Ułamkowe, 34, 344, 350

formatowanie, 559
daty i czasu, 352
komórek, 340

paski danych, 367
pustych, 373
skala barw, 365
zakresu, 365, 367, 369
zestaw ikon, 369
zawierających tekst, 372
zawierających wartości błędu, 373

liczb dziesiętnych, 374
liczby, daty lub godziny, 373
pobranych danych, 481
sekcji, 344
wartości

powyżej średniej, 376
sklasyfikowanych jako pierwsze, 375
unikatowych, 378
zduplikowanych, 378

warunkowe, 275, 361
opcje, 362, 364
reguły, 385
ustalanie zakresu, 381
usuwanie, 384
zamiana na stałe, 384
znajdowanie komórek, 386

według daty, 373
według tekstu, 373

formularz, 389
formuła, 32, 39, 560

„zatrzaskowa”, 560
dynamiczna, 580
tablicowa, 39, 108, 130, 169, 206

edycja, 176
kopiowanie, 178
przekształcanie w formułę zwykłą, 178
zamiana wyników, 177
zmiana zakresu, 176

zwykła, 39, 178

formuły
autouzupełnianie, 72
dla komórek do sformatowania, 380
kopiowanie, 157, 162
łączenie, 169
maskowanie błędów, 282
modyfikacja adresów, 160
przenoszenie komórek, 173
sprawdzanie poprawności, 292
szacowanie, 296
ustalanie rankingu, 456
w komórkach scalonych, 167
wprowadzanie adresów, 42
wprowadzanie funkcji, 71
z odwołaniami względnymi, 163
zapis w Excelu, 75
zapis w matematyce, 75
zawierające nazwy, 162

FTP, File Transfer Protocol, 497
funkcja, 561

ADR.POŚR, 45, 92, 96, 138, 181
ADRES, 90
AGREGUJ, 133, 422
CZĘSTOŚĆ, 144, 325, 328, 462, 468
CZY.BŁ, 278
CZY.BŁĄD, 278
CZY.BRAK, 278
DZIŚ, 181
HIPERŁĄCZE, 502, 506, 508

z argumentem tablicowym, 507
ILE.LICZB, 227, 336
ILE.NIEPUSTYCH, 227
ILE.WIERSZY, 90, 144
INDEKS, 138, 185, 187, 189, 208, 252, 318

forma odwołaniowa, 189
INFO, 103, 181
JEŻELI, 82, 84, 123, 125, 129, 130

wielokrotne zagnieżdżenie, 190
JEŻELI zwracająca wartość pustą, 83
JEŻELI.BŁĄD, 87, 112, 144, 282, 290, 337
JEŻELI.ND, 112
KOMÓRKA, 57, 100–103, 181, 359
L, 99
LICZ.JEŻELI, 143, 281, 429, 433, 437, 464
LICZ.WARUNKI, 240, 435
LICZBA.CAŁK, 79
LICZBA.KOLUMN, 90
LOS, 181

LOS.ZAKR, 181
MACIERZ.ILOCZYN, 117, 120, 122, 202
MAKS.WARUNKÓW, 440
MAX, 208, 451
MAX.A, 225
MAX.K, 454
MIN.K, 454
MOD, 79, 382
MODUŁ.LICZBY, 78
NR.BŁĘDU, 280
NR.KOLUMNY, 88, 450
OBLICZ, 303
PIERWIASTEK, 78
PODAJ.POZYCJĘ, 97, 98, 318, 321, 327, 472
POTĘGA, 78
POZYCJA, 145
PRZEŁĄCZ, 86
PRZESUNIĘCIE, 57, 138, 181, 191, 194, 208
RADIANY, 77
RANDBETWEEN, 181
SILNIA, 239
STOPNIE, 77
SUMA.ILOCZYNÓW, 81, 252
SUMA.JEŻELI, 143
SUMA.WARUNKÓW, 240, 435
SUMA.X2.M.Y2, 81
SUMA.X2.P.Y2, 81
SUMA.XMY.2, 81
SUMY.CZĘŚCIOWE, 133, 419–421
SUMY.KWADRATÓW, 81
ŚREDNIA.JEŻELI, 240, 436
ŚREDNIA.WARUNKÓW, 240, 436
T, 99
TERAZ, 181
TRANSPONUJ, 117–120
TYP, 279
WARUNKI, 85
WIERSZ, 88, 197, 336, 382, 450
WYBIERZ, 83, 84, 123, 126–130
WYST.NAJCZĘŚCIEJ, 141
WYSZUKAJ, 315, 318, 467

składnia tablicowa, 318
składnia wektorowa, 315

WYSZUKAJ.PIONOWO, 311, 314
WYSZUKAJ.POZIOMO, 311
XOR, 233
ZAOKR, 78
ZAOKR.DO.CAŁK, 79

funkcje
agregujące, 215, 249

nietypowe, 217
aproksymacja, 555
argumenty, 77

opcjonalne, 227, 229
puste, 227

autouzupełniania formuł, 136
baz danych, 444, 447
bliźniacze, 225
cyklometryczne, 357, 358
do diagnozowania błędów obliczeń, 278
do obliczania logarytmów, 78
do odzyskiwania danych, 409
do sumowania, 81
do wykonywania działań na macierzach, 116
do zaokrąglania liczb, 78
dwu zmiennych, 304
ekspansywne, 113, 223
grupowanie, 213
interpretacja

argumentów, 215
danych grupowych, 254
wyrażeń złożonych, 220

inżynierskie, 223, 248
jednej zmiennej, 299, 310
kolejność danych, 219
liczba zwracanych wyników, 218, 223
logiczne, 85
matematyczne, 75
minimalna liczba argumentów, 216
odwołania 3-W, 141, 221, 240
parametryczne, 304
podsumowań, 419
pomijanie argumentów, 227
redukujące, 223
rodzaje argumentów, 214
specjalne, 99, 222
statystyczne, 227
tablicowe, 218, 249, 299, 304
translacyjne, 223
trygonometryczne, 76, 357
ulotne, 181
ułatwiające konwersję typów, 234
użycie argumentów innego typu, 236
w polu obliczeniowym, 487
wbudowane, 509–52
wywołania, 71

zagnieżdżanie, 71
zliczania i sumowania warunkowego, 426
zwracające odwołania, 224
zwykłe, 222, 248

H
hiperłącza, 497

automatyczne tworzenie, 500
puste, 507
tworzenie, 498
usuwanie, 500
warunkowe, 507

hiperłącze
do komórki, 503, 505
do określonego obszaru, 504
do określonej lokalizacji, 504
do programu pocztowego, 508
do skoroszytu, 503
do skoroszytu na innym dysku, 504
do skoroszytu na serwerze, 504
do zakresu arkusza, 504

HTTP, Hypertext Transfer Protocol, 497

I
iloczyn zbiorów, 469
importowanie danych, 479
informacje o formacie, 359
inspekcja formuł, 264
interpolacja, 310
interpretacja

adresów, 247, 251
danych grupowych, 254
nazw, 247, 251

J
język VBA, 24

K
klawisz funkcyjny F9, 153
kody

błędów, 278
formatu, 341, 356

daty i czasu, 355
komórek, 339
w funkcjach, 357

kolejność
działań, 70
sortowania, 390

kolor tekstu, 342
kolumny, 31

dodawanie, 137
liczby w nagłówkach, 139
obliczeniowe, 132
pomocnicze, 290
przestawianie, 314
ukrywanie nagłówków, 137
usuwanie, 137, 174
wstawianie, 174
zmienianie nazwy, 137

komentarz, 562
komórki, 31, 562

informowanie o formacie, 359
o wyliczonym adresie, 188
puste, 32, 287

zliczanie, 441
rodzaje formatowania, 340
scalone, 44, 58, 164

formuły, 167
formuły tablicowe, 169

z hiperłączem, 501
zależne, 293

konfiguracja listy rozwijanej, 261
koniunkcja kryteriów, 403
konsolidacja, 414

według kategorii, 417
według położenia, 414

kontrola
błędów, 278
poprawności wprowadzania danych, 257

konwersja
na tekst, 236
na wartość liczbową, 237
na wartość logiczną, 236
typów argumentów, 231
typów danych, 234
zakresu na tabelę, 140

kopiowanie
formuł, 157, 162, 167, 169
formuł tablicowych, 178
komórek widocznych, 170
odwołań, 162
za pomocą schowka, 171

Kreator kwerend, 477

kryteria, 428, 432, 482, 563
alternatywa, 403
koniunkcja, 403
określanie rekordów, 482
sortowania, 458
wyrażenia logiczne, 404
z wartościami błędów, 429

kwalifikator, 47
nazwy arkusza, 245

kwerendy, 477
miejsce docelowe wyników, 480
parametryczne, 481, 489
pobieranie tabel, 494
ponowne używanie, 479
udostępnianie, 479

L
liczba, 37

całkowita, 239
liczbotekst, 563
licznik
generujący liczbę numerów, 452

oparty
na funkcji MAX, 451
na funkcji WIERSZ, 450
na odwołaniu względnym, 449

statyczny, 449
z filtrem, 453

lista
niestandardowa, 564
rozwijana, 564

aparat fotograficzny, 275
dynamiczna, 265, 270
jako podpowiedź, 263
kody formatu, 349
konfiguracja, 261
tworzona ad hoc, 263
wynik wyboru, 274
z formatowanie warunkowym, 275
z obsługą zdarzenia Change, 276

unikatów, 464–466
wyboru

z aktywną podpowiedzią, 271
zwracająca dane

liczbowe, 349
tekstowe, 350

zależna, 272

literał, 39
tablicowy, 106

logarytm, 78
Lotus, 29

Ł
łącze, link, 46, 120, 564
łączenie

formuł i stałych, 169
fragmentów tablic, 189
tablic, 114, 203

M
macierz, 105, 574

diagonalna, 106
jednostkowa, 121
kwadratowa, 106
odwrotna, 121
transponowana, 106
trójkątna, 122

macierze
wyznacznik, 121

maksimum warunkowe, 440
maskowanie wartości błędów, 281, 283

metody, 284
Menedżer

nazw, 57, 63
reguł formatowania warunkowego, 365, 372,

376
metoda Evaluate, 302
Microsoft Query, 475

definiowanie kwerendy, 477
działanie programu, 475
okienka kryteriów, 483
określanie zakresu wartości, 484
wyszukiwanie rekordów, 483

minimum warunkowe, 440
mnożenie macierzowe, 116

N
nadawanie nazw, 54
nagłówki kolumn, 139
nawiasy

klamrowe, 107, 575
kwadratowe, 46
okrągłe, 72

nazwa, 564
a odwołania cykliczne, 60
automatyczna wymiana adresów, 62
formuł, 51, 59
komórek, 51
komórek scalonych, 58
kwalifikowane, 58
lokalne, 58
metody nadawania, 54
modyfikacja, 61
obszarów, 51
odczyt, 102
odwołań wielopoziomowych, 142
stałych, 51, 59
tabel, 133
tablic, 130
wyświetlanie listy, 63
z zaznaczenia, 55
zakres obowiązywania, 53
zakresów

a identyfikatory kolumn, 57
komórek, 53
wierszowych i kolumnowych, 55

zalety i wady, 63
nieprawidłowe dane, 264
numerowanie kolumn z danymi, 449

O
obliczanie

logarytmów, 78
reszty z dzielenia, 79
wartości kryteriów, 485

obsługa
baz danych, 475
długich ciągów cyfr, 437

obszar, 41, 105, 565
przyległy, 566
roboczy, 566

odczyt
adresów, 88
gęstości roztworu, 330
kodu formatu, 356
nazwy, 102

odświeżanie danych zewnętrznych, 482
odwołania, 41, 564, 566

cykliczne, 49, 60, 566
do danych zewnętrznych, 46
do elementów tabeli, 138

do elementów tablic, 185
do fragmentów tablic, 185
do komórki, 188
do zamkniętych skoroszytów, 240
mieszane, 95
strukturalne, 131, 567

kopiowanie, 137
kwalifikowane, 135
niekwalifikowane, 135
przenoszenie, 137
używanie nazw, 139
wypełnianie, 137
zewnętrzne, 136

wielopoziomowe, 3-W, 49, 140, 567
a zmiany w arkuszu, 143
nadawanie nazwy, 142
zastosowania, 143

względne, 95, 449
zewnętrzne, 96

odwracanie porządku elementów
w tablicy, 190
w zakresie, 195

odzyskiwanie danych, 409
okno inspekcji kwerendy, 480
określanie zakresu wartości, 484
OneDrive, 27
opcje

autokorekty, 501
automatycznego formatowania, 501
formatowania warunkowego, 362, 364

operacje
konsolidacji, 414
tablicowe, 110
transpozycji, 119

operator, 567
Between, 489
In, 489
Is Nie Null, 489
Is Null, 488
Like, 488

operatory
adresowe, 65
arytmetyczne, 231, 485
działań, 64
jednoargumentowe, 233
logiczne, 488
odwołania, 134
porównania, 485
relacji, 67

P
parametr, 568
paski danych, 367
permutacja, 266
pierwszeństwo wykonywania działań, 64
plan eksperymentu czynnikowego, 200
pobieranie

rekordów, 490
tabel, 494

podzbiór, 582
pole, 389, 573

kombi, 276
listy, 564
nazwy, 53
obliczeniowe, 486

funkcje, 487
połączenie ze źródłem danych, 476
poprawność

danych, 264
formuł, 292

porównywanie
danych różnych typów, 69
tekstów, 68
zbiorów danych, 469

priorytet operatorów, 64
program

Lotus, 29
Microsoft Query, 475

protokół
FTP, 497
HTTP, 497

przeliczanie
arkusza, 179
automatyczne, 180
automatyczne z wyjątkiem tabel danych, 180
ręczne, 180
skoroszytu, 179

przenoszenie
komórek, 173
powiązanych skoroszytów, 174

przestawianie wierszy, 314
przeszukiwanie tablic, 310
przyczyny błędów, 294
przyspieszanie obliczeń, 292
pseudofunkcja, 569
pseudotekst, 569
pseudozakres, 41, 569

puste
argumenty, 227, 228
hiperłącza, 507
komórki, 32, 228, 287
komórki jako dane, 288
teksty, 287

usuwanie, 289

R
ranking danych, 453
regresja, 570
reguły formatowania warunkowego, 385
rekord, 389, 573
rekurencja, 570
relacje, 67
reprezentacja tablic, 106
rozmiary obszaru formuły, 206
różnica zbiorów, 470

S
samosprzężenia, 491
scalanie komórek, 165

niestandardowe, 167
standardowe, 165

schowek, 171
sekwencje

językowe, 356
warunkowe, 343

seria odwołań bezwzględnych, 161
skala barw, 365
skalar, 106
składnia

tablicowa funkcji WYSZUKAJ, 318
wektorowa funkcji WYSZUKAJ, 315

skoroszyt, 31, 494
wstawianie hiperłącza, 499

skrót klawiaturowy
Ctrl+Alt+F9, 180
Ctrl+Alt+V, 120, 159
Ctrl+C, 170
Ctrl+D, 159
Ctrl+Enter, 159
Ctrl+F, 156
Ctrl+H, 156
Ctrl+Shift+Alt+F9, 180
Ctrl+Shift+Enter, 39, 108, 211
Shift+F9, 180

sortowanie, 390, 392, 458
bazy danych, 390
zakresów

odwołania do komórek, 397
zawierających formuły, 396

spacja, 73, 139, 346
specjalne wyrażenie indeksowe, SWI, 572
specyfikator

tabeli, 133
wierszy, 134

sprzęganie tabel, 490
sprzężenia
różnicowe, 493

wewnętrzne, 490
zewnętrzne, 490, 491, 494

pełne, 492
stała, 572

liczbowa, 32, 36, 39
tablicowa, 106, 248, 575
tekstowa, 33, 39
logiczna, 33, 39

stężenie procentowe, 330
struktura

adresu URL, 497
arkusza kalkulacyjnego, 31
kodów formatu, 341
tabeli bazy danych, 389
tablicy, 210
wyrażeń, 209

styl adresowania, 40, 506, 573
A1, 95, 506
W1K1, 91, 506

suma
częściowa, 422
zbiorów, 471

sumowanie, 81
warunkowe, 426, 438

superpozycja, 112
sygnalizacja błędów, 278
symbole funkcji matematycznych, 76
szacowanie formuł, 296

Ś
śledzenie komórki, 103
średnik, 114, 341, 561

T
tabele, 389, 573

danych, 308, 448
elementy składowe, 134
formuły tablicowe, 139
nazwa, 133
odwołania strukturalne, 131
pełne sprzężenia zewnętrzne, 492
przestawne, 574
samosprzężenia, 491
specyfikator, 133
sprzężenia

różnicowe, 493
wewnętrzne, 491
zewnętrzne, 491, 494

strukturalne, 131
zmienianie nazwy, 137

tablica pierwiastków chemicznych, 393
tablice, 105, 574

bazowe, 197, 576
definiowanie innych tablic, 199

do obliczeń, TDO, 575
do wyświetlania, TDW, 142, 213, 218, 575
dynamiczne generowanie, 197
dwuwymiarowe, 201
jednowymiarowe, 197
listowanie zawartości, 191
łączenie, 203
łączenie fragmentów, 189
mieszane, 114
nazwane, TN, 213, 576
nazwy formuł, 248
nieuporządkowane, 317
odwracanie porządku elementów, 190
pełnowartościowe, TPW, 204, 212, 576
quasi-pełnowartościowe, TQP, 215, 576
selektywne sumowanie elementów, 201
struktura, 210
wyszukiwanie danych, 336

liczbowych, 329
znaczących, 331

z pamięcią adresu, TPA, 124, 576
tablicowanie funkcji

dwu zmiennych, 304
jednej zmiennej, 299

tekst, 33, 37
określenie koloru, 342

tekst adresu, 41
transpozycja, 117, 118, 120

danych zduplikowanych, 472
tworzenie

hiperłącza, 498
kwerendy parametrycznej, 489
list dynamicznych, 269
list zależnych, 272
listy unikatów, 464
nazw z zaznaczenia, 55
odwołań wielopoziomowych, 142
pola obliczeniowego, 486
serii odwołań, 161
tablic mieszanych, 114

typy
danych, 32, 577
informacji, 101, 104

U
uchwyt wypełniania, 158
układ dwuwymiarowy, 333
ukośnik wsteczny, 107
ukrywanie

nagłówków kolumn, 137
wartości zerowych, 285

ulotny, 577
ułamek, 239

złożony, 76
unikaty, 464, 465, 466
URL, Uniform Resource Locator, 497
usuwanie

błędnych danych, 290
formatów warunkowych, 384
hiperłącza, 500
kolumn, 174
komórek, 174
pustych tekstów, 289
wierszy, 174

użycie
adresów względnych i mieszanych, 242
autofiltra, 399
dynamicznej listy rozwijanej, 266, 268, 270
filtra zaawansowanego, 406
formuł tablicowych, 130, 139
formuł zamiast filtra, 407
klawisza F9, 153
listy dynamicznej, 266

594 Zrozumieć Excela. Funkcje i wyrażenia

użycie
nazw tablic, 130
odwołań strukturalnych, 135, 136
pasków danych, 367
plusa jednoargumentowego, 233
sekwencji językowych, 356
skali barw, 365
znaku spacji, 139

V
VBA, Visual Basic for Applications, 24, 578

W
wartości

błędów, 33
domyślne, 227
logiczne, 33, 231
unikatowe, 459, 463
zerowe, 285

wektor, 105
wersja online

ograniczenia funkcjonalności, 27
wiersz, 31
wklejanie

łącza, 120, 499
specjalne, 159, 160

wprowadzanie
adresów do formuł, 42
danych, 257
funkcji do formuł, 71

wskaźnik położenia, 454
wstawianie

kolumn, 174
komórek, 174
wierszy, 174

wykres, 578
przestawny, 579

wykrzyknik, 46, 245, 246
wyliczanie adresów, 88
wyniki kwerendy, 480, 483
wyrażenia, 579, Patrz także formuła

adresy, 250
bazujące na wartościach odwołań, 247
dynamiczne, 580
interpretacja nazw, 247
jednowartościowe, 209

logiczne, 404
błędów, 283

nazwy, 250
proste, 212, 580
tablicowe, 123

interpretacja nazw, 251
wielowartościowe, 209
z odwołaniami, 213
z tablicami nazwanymi, 213
złożone, 213, 580

wyróżnianie wierszy, 383
wyszukiwanie

cyfr, 313
danych, 312

liczbowych, 329
w tablicy, 336
w układzie dwuwymiarowym, 333
w zakresie, 334
zduplikowanych, 326

liczby, 319
odpowiedników wartości, 322
rekordów, 484
wartości, 322
wartości wspólnych, 469
wszystkich wystąpień, 323
znaczących danych, 331

wyświetlanie
danych, 147
tekstów formuł, 148

wywołania funkcji, 71
wyznacznik macierzy, 106, 121
wzorce formatu sekcji, 344

Z
zagnieżdżanie funkcji, 71
zakres, 41, 105, 580

do wyświetlania, ZDW, 218, 575
dynamiczny, 204, 580

adres początku i końca, 205
długość, 205
początek obszaru, 205
stały początek obszaru, 205

formatowania, 381
komórek, 365, 367, 369

kontrolny, 432
nieregularny, 219
odwracanie porządku elementów, 195

  Skorowidz 595

sumowania, 432
złożony, 66
znajdowanie wartości, 317, 334

zakresy
część wspólna, 55, 196
wierszowe i kolumnowe, 55

zamiana wyników formuł, 177
zaokrąglanie liczb, 78
zapis naukowy, 34
zasada

rzutu prostokątnego, 44, 247, 581
sekwencji, 581
superpozycji, 112, 581

zasięg, 581
zatwierdzenie tablicowe, 211
zaznaczanie

całego arkusza, 150
kolumn, 150
komórek, 149
nieprzylegających komórek, 149
ostatnich komórek arkusza, 151
wierszy, 150
zakresów komórek, 151

zbiory, 582
iloczyn, 469
różnica, 470
suma, 471

zdarzenie Change, 276
zliczanie

danych, 326
pustych komórek, 441
wartości unikatowych, 459
warunkowe, 426, 438

zmienianie
czcionki, 182
nazwy tabeli, 137
zakresu formuły, 176

zmienna, 582
znajdowanie i zamiana, 156
znak

#, 345
%, 346
*, 346
,, 345, 346, 359
., 345
/, 346
?, 345
!, 46, 245, 246
@, 341, 347
;, 114, 341, 561
0, 345
C, 359
D, 359
E–, 346
F, 359
G, 359
H, 359
P, 359
S, 359
spacji, 73, 139, 346

znaki wieloznaczne, 313

Ź
źródło danych, 476

Notatki

